

INGENIERÍA TÉCNICA INDUSTRIAL - Esp. Mecánica

FUNDAMENTOS MATEMÁTICOS DE LA INGENIERÍA**Práctica nº 5: Funciones de una variable**

RESUMEN: Abordamos en esta práctica el tratamiento con MAPLE de las funciones en una variable, así como el cálculo de límites y el estudio de la continuidad de dichas funciones.

COMANDOS MÁS IMPORTANTES:

with(plots), display, plot, limit, Limit

- Introducción y recordatorio

Recordamos que se pueden representar en unos mismos ejes varias gráficas a la vez. Por ejemplo dos funciones, o bien, una función y una serie de puntos. El primer caso se ha estudiado bastante y lo realizamos por medio de la orden es:

plot({función1,función2}, x=x0..x1, y=y0..y1, title=`título de la gráfica`, color=[color 1,color 2], thickness=[grosor 1,grosor 2], linestyle=[tipo1,tipo 2], discount=true)

En el segundo caso estaríamos haciendo dos gráficas con diferentes formatos por lo que hay que recurrir a la opción del display que se encuentra en el paquete plots para visualizarlas en unos mismos ejes coordenados. Para hacer una gráfica de este tipo primero se carga el paquete específico para gráficos de MAPLE, utilizando la orden

with(plots):

A continuación se definen las gráficas que se quieren dibujar

Grafica1:= plot(f1(x), x=a..b, opciones):

Grafica2:=plot(f2(x), x=a..b, opciones):

Es aconsejable terminar estas órdenes con dos puntos. Si se termina con punto y coma, el resultado será un listado de puntos. Finalmente se utiliza la orden **display** para superponer los dos gráficos.

display(Grafica1,Grafica2);

En la ayuda del **plot** se pueden ver diversas opciones. A continuación citaremos dos que no se han visto:

axes=framed/boxed/normal/none

Especifica el tipo de ejes.

$labels=[x,y]$

Especifica etiquetas para los ejes

Ejemplo: Dibujar en unos mismos ejes las funciones $f(x) = \sin(x)$ y $g(x) = x$.

Solución 1:

```
[ > plot({sin(x), x}, x=-20..20, labels=[etiqueta1, etiqueta2]);
```

Solución 2:

```
[ > with(plots) :  
[ > grafica1:=plot(sin(x), x=-6*Pi..6*Pi) :  
[ > grafica2:=plot(x,x=-20..20) :  
[ > display({grafica1,grafica2}, style=point);
```

Si no se va a usar ninguna opción se puede hacer de la siguiente forma

```
[ > display(grafica1,grafica2);
```

- Dibujo de un conjunto de puntos y poligonales

En muchas ocasiones es muy útil dibujar en una gráfica puntos aislados o líneas rectas aisladas o conectadas entre sí formando líneas poligonales.

Una lista en Maple es una secuencia de números o expresiones separadas por comas y encerradas entre corchetes. Los puntos en Maple son listas de elementos (las coordenadas del punto). Así por ejemplo, el punto de coordenadas $x=1$, $y=1$, en Maple se escribe $[1,1]$.

Para la representación gráfica de puntos, los argumentos obligatorios en la orden **plot** son: la lista de puntos especificando sus coordenadas (entre corchetes) y la expresión **style=point**.

plot([[a1,b1],[a2,b2],[a3,b3],...], x=a..b, y=c..d, style=point)

En este caso no es necesario poner la opción del rango de los ejes.

Se puede elegir el símbolo a utilizar para la representación de los puntos mediante la opción **symbol=tipo de símbolo**. Entre los tipos de símbolos están los siguientes: *box*, *cross*, *circle* y *diamond*.

El omitir el estilo en la orden anterior produce la poligonal que une dichos puntos, esto es la unión de segmentos que une los puntos.

Ejemplo: Representar los puntos $[0,0]$, $[1,1]$, $[2,0]$, $[-2,1]$, $[-1,1]$, $[0.5, \sqrt{2}]$.

Solución:

```
[ > plot([[0,0],[1,1],[2,0],[-2,1],[-1,1],[0.5,sqrt(2)]],  
style=point, symbol=box, color=blue);
```

Ejemplo: Representar la poligonal que une los puntos $[0,0]$, $[1,1]$, $[2,0]$, $[-2,1]$, $[-1,1]$, $[0.5, \sqrt{2}]$.

Solución:

```
[ > plot([[0,0],[1,1],[2,0],[-2,1],[-1,1],[0.5,sqrt(2)]],  
color=red);
```

- Curvas en paramétricas

En muchos problemas físicos las funciones en el plano se suelen describir mediante un parámetro, que normalmente es el tiempo. Con el fin de motivar la representación en

paramétricas, veamos un ejemplo.

Consideremos el camino que sigue un objeto cuando lo lanzamos hacia arriba con un determinado ángulo y que supongamos que viene dado por el camino parabólico dado por

$$(1) \quad y = -\frac{x^2}{4} + x$$

Si lo dibujamos tenemos

```
[ > plot(-x^2/4+x, x=0..4, y=0..1.5);
```

Sin embargo esta ecuación no nos cuenta toda la historia. Es decir nos describe el movimiento global del objeto pero no nos dice en qué punto (x,y) está el objeto en un instante de tiempo t.

Para ello se introduce una variable llamada parámetro. Si escribimos los puntos (x,y) en función de un parámetro t por ejemplo de la forma

$$x = 2t$$

$$y = -t^2 + 2t$$

obtenemos las ecuaciones paramétricas de la función dada en (1).

En general, se dice que una curva viene dada en forma paramétrica si está dada en la forma

$$x = x(t)$$

$$y = y(t)$$

donde t es un parámetro que varía en algún intervalo, por ejemplo [a,b], o incluso en todo los números reales.

Para representarla usamos el comando

plot([x(t), y(t), t=a..b], opciones)

El hecho de que la lista conste de tres elementos y el tercero de ellos sea el rango, hace que el programa reconozca una gráfica en paramétricas en lugar de un gráfico múltiple. Una opción útil es la de

scaling= constrained

Esta opción lo que nos permite es que el dibujo no lo deforme, es decir que la escala sea la misma 1:1. Maple usa por defecto la opción **unconstrained**.

Ejemplo: Representar la curva en paramétricas:

$$x = 2(t - \sin t)$$

$$y = 2(1 - \cos t)$$

```
[ > plot([2*(t-sin(t)), 2*(1-cos(t)), t=0..2*Pi], scaling=constrained);
```

Pueden combinarse en un gráfico múltiple curvas en cartesianas y en paramétricas.

Ejemplo: Representar en unos mismos ejes las funciones $x(t) = 3 \cos(t)$, $y(t) = \sin(t)$, $f1(x)=x$ y $f2(x)=-x$.

Solución:

```
[ > plot([3*cos(t), sin(t), t=0..2*Pi], x, -x), x=-4..4, scaling=constrained, color=[red,black,blue]);
```

A veces la representación en paramétricas no permite hacer el dibujo en todo los números reales, por ejemplo una recta, para evitarlo hay que limitar el rango del parámetro.

Veamos algunos ejemplos importantes de funciones en paramétricas.

Ejemplo: Representar la **recta** que pasa por el punto $P=(8,3)$ y tiene como vector director $v=(1,-1)$.

Solución: La ecuación de la recta es $x=8+t$, $y=3-t$.

```
[ > plot([8+t,3-t,t=-infinity..infinity], title=`Gráfica  
  utilizando todo los números reales`);  
[ > plot([8+t,3-t,t=-20..20], title=`Gráfica de la recta que pasa  
  por P=(8,3) con dirección v=(1,-1)`, scaling=constrained);
```

Ejemplo: Representar la **recta** que pasa por el punto $P=(4,-5)$ y tiene como vector director $v=(2,-3)$.

Solución: La ecuación de la recta es $x=4+2t$, $y=-5-3t$.

```
[ > plot([4+2*t,-5-3*t,t=-20..20], title=`Gráfica de la recta que  
  pasa por P=(4,-5) con dirección  
  v=(2,-3)`, scaling=constrained);
```

Ejemplo: Representa en paramétricas la ecuación de la **circunferencia** de centro $(1,-1)$ y radio $\sqrt{2}$

Solución: La ecuación de la circunferencia es

$$x = 1 + \sqrt{2} \cos(t)$$
$$y = -1 + \sqrt{2} \sin(t)$$

donde t es un parámetro en un intervalo de longitud 2π , por ejemplo $[0, 2\pi]$, $[-\pi, \pi]$ etc...

Si hacemos el dibujo sin la opción `scaling=constrained`, nos sale bastante deformado, de hecho aparenta una elipse.

```
[ > plot([1+sqrt(2)*cos(t),-1+sqrt(2)*sin(t),t=-Pi..Pi],  
  title=`Representación de una circunferencia de centro (1,-1)  
  y radio sqrt(2), sin la opción scaling`);  
[ > plot([1+sqrt(2)*cos(t),-1+sqrt(2)*sin(t),t=-Pi..Pi],  
  scaling=constrained, title=`Representación de una  
  circunferencia de centro (1,-1) y radio sqrt(2)`);
```

¿Qué ocurre si cambiamos el rango del parámetro t , por ejemplo $[0, \pi]$, $\left[0, \frac{\pi}{4}\right]$?

```
[ > plot([1+sqrt(2)*cos(t),-1+sqrt(2)*sin(t),t=0..Pi],  
  scaling=constrained);  
[ > plot([1+sqrt(2)*cos(t),-1+sqrt(2)*sin(t),t=0..Pi/4],  
  scaling=constrained);
```

Ejemplo: Dibujar la **elipse** de centro $(0,0)$ con semieje X igual a 4 y semieje Y igual a 2.

Solución: La ecuación de la elipse es:

$$x = 4 \cos(t)$$
$$y = 2 \sin(t)$$

donde t se mueve en cualquier intervalo de longitud 2π . Igual que en la circunferencia, para que no salga deformada se usa la opción `constrained`:

```
[ > plot([4*cos(t),2*sin(t),t=0..2*Pi], scaling=constrained);
```

Ejemplo: El cardioides es una curva de una forma especial, que en general tiene por ecuaciones

$$x(t) = (a + b \cos(t)) \cos(t)$$

$$y(t) = (a + b \cos(t)) \sin(t)$$

donde a y b son valores reales fijos. ¿En qué intervalo de la variable t queda definida completamente esta curva?

Dibujar el cardioides con a=b=1. Modificar los valores de a y b y representarlo

```
[ > plot([(1+cos(t))*cos(t), (1+cos(t))*sin(t), t=0..2*Pi], title='C  
ardioides', scaling=constrained);  
[ > plot([(cos(t))*cos(t), (2+cos(t))*sin(t), t=0..2*Pi], title='Car  
dioides', scaling=constrained);  
[ > plot([(2+cos(t))*cos(t), (cos(t))*sin(t), t=0..2*Pi], title='Car  
dioides', scaling=constrained);
```

Ejemplo: Se llama cicloide a toda curva plana que se obtiene físicamente por la trayectoria de un punto de una circunferencia que, sin deslizarse rueda por una recta horizontal. La ecuación en paramétricas de una cicloide es:

$$x(t) = a(t - \sin(t))$$

$$y(t) = a(1 - \cos(t))$$

Representar la cicloide con parámetro a=1 y 3.

```
[ > plot([t-sin(t), 1-cos(t), t=0..6*Pi], scaling=constrained);  
[ > plot([3*(t-sin(t)), 3*(1-cos(t)), t=0..6*Pi], scaling=constrained);
```

- Límite de una función en un punto

Este es el concepto fundamental del cálculo infinitesimal.

La sentencia que usa MAPLE para calcular el límite de una función f(x) en un punto a es la misma que ya vimos para estudiar el límite de una sucesión numérica:

$$\mathbf{\text{limit}(f(x), x=a, dir)}$$

donde *dir* es opcional e indica la dirección en el caso de límites laterales: *left* (hacia la izquierda) o *right* (hacia la derecha). Si se omite se calcula el límite global, que existe si y sólo si existen los dos límites laterales y son iguales.

Si queremos que nos aparezca antes del valor la expresión matemática

$$\lim_{x \rightarrow a} f(x)$$

tenemos que usar la sentencia **Limit** de la siguiente forma

$$\mathbf{\text{Limit}(f(x), x=a, dir) = \text{limit}(f(x), x=a, dir)}$$

Ejemplo: Calcular los límites siguientes:

$$(a) \lim_{x \rightarrow 0} \frac{\text{sen}(x)}{x} \quad (b) \lim_{x \rightarrow 1} \frac{|x^2 - 1|}{x - 1} \quad (c) \lim_{x \rightarrow 0} \frac{1}{x}$$

```
[ > Limit(sin(x)/x, x=0)=limit(sin(x)/x, x=0);
```

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x} = 1$$

> `f:=x->abs(x^2-1)/(x-1);`

$$f := x \rightarrow \frac{|x^2 - 1|}{x - 1}$$

> `Limit(f(x), x=1)=limit(f(x), x=1);`

$$\lim_{x \rightarrow 1} \frac{|x^2 - 1|}{x - 1} = \text{undefined}$$

En vista de esta indefinición, estudiamos los límites laterales:

> `Limit(f(x), x=1, left)=limit(f(x), x=1, left);`

$$\lim_{x \rightarrow 1^-} \frac{|x^2 - 1|}{x - 1} = -2$$

> `Limit(f(x), x=1, right)=limit(f(x), x=1, right);`

$$\lim_{x \rightarrow 1^+} \frac{|x^2 - 1|}{x - 1} = 2$$

Por tanto, como los límites laterales son distintos, no existe el límite y por eso arriba aparece la palabra *undefined*.

> `Limit(1/x, x=0)=limit(1/x, x=0);`

> `Limit(1/x, x=0, left)=limit(1/x, x=0, left);`

> `Limit(1/x, x=0, right)=limit(1/x, x=0, right);`

Continuidad de una función

Una función $f(x)$ se dice continua en un punto a si $\lim_{x \rightarrow a} f(x) = f(a)$. En caso contrario se dice que es discontinua en dicho punto.

Por tanto, para que una función sea continua en a se debe verificar:

- 1.- La función $f(x)$ debe estar bien definida en $x=a$.
- 2.- Tiene que existir (y ser finito) el límite $\lim_{x \rightarrow a} f(x)$
- 3.- El valor de dicho límite debe coincidir con $f(a)$.

Ejemplo: Estudiar la continuidad de la función

$$f(x) = 1 \text{ si } x = 0 \text{ y } f(x) = \frac{1}{1 + e^{\left(\frac{1}{x}\right)}} \text{ en el resto.}$$

Por definición, $f(0)=1$, por lo que tenemos que hallar el límite de $f(x)$ cuando x tiende a 0:

> `limit(1/(1+exp(1/x)), x=0);`

undefined

Como no hay límite global, estudiamos los laterales:

> `limit(1/(1+exp(1/x)), x=0, left);`

1

> `limit(1/(1+exp(1/x)), x=0, right);`

0

Por tanto, no existe límite cuando x tiende a 0, por lo que $f(x)$ es discontinua en 0. Para el resto de valores de x , $f(x)$ es continua pues es un cociente de funciones continuas cuyo denominador no se anula.

Ejemplo: Estudiar la continuidad en $x=0$ de la función $f(x) = \frac{e^x}{1+e^x}$ si $x \leq 0$ y $f(x) = \frac{x^2+1}{2}$ si $0 < x$.

Como en este caso la expresión de $f(x)$ varía dependiendo de si está a la derecha o a la izquierda del 0, no podemos usar directamente la sentencia **limit** global de MAPLE, pues esta supone que la expresión de $f(x)$ es la misma por ambos lados. Para este caso hay que utilizar los límites laterales. **IMPORTANTE:** ¡no confundir los lados!:

```
> fiz:=x->exp(x)/(1+exp(x)); fder:=x->(x^2+1)/2;
```

$$fiz := x \rightarrow \frac{e^x}{1+e^x}$$

$$fder := x \rightarrow \frac{1}{2}x^2 + \frac{1}{2}$$

Valor de $f(0)$:

```
> fiz(0);
```

$$\frac{1}{2}$$

Límites laterales:

```
> Limit(fiz(x), x=0, left)=limit(fiz(x), x=0, left);
```

$$\lim_{x \rightarrow 0^-} \frac{e^x}{1+e^x} = \frac{1}{2}$$

```
> Limit(fder(x), x=0, right)=limit(fder(x), x=0, right);
```

$$\lim_{x \rightarrow 0^+} \frac{1}{2}x^2 + \frac{1}{2} = \frac{1}{2}$$

Como coinciden los límites laterales con el valor de $f(0)$, f es continua en el 0.


Para estudiar la continuidad, también es muy útil usar la **representación gráfica** de las funciones. En el ejemplo anterior podríamos representar las funciones y ver qué pasa. Cuando tenemos funciones definidas a trozos, el valor de x varía de forma diferente en un trozo que en otro por lo que hay que usar **display**, cargando previamente el paquete **plots**:

```
[ > with(plots) :
```

```
[ > g1:=plot(fiz(x), x=-1..0, color=green) :
```

```
[ > g2:=plot(fiz(x), x=0..1, color=blue) :
```

```
[ > display({g1,g2}) ;
```



También la representación gráfica es muy útil para detectar los posibles puntos de discontinuidad de una función.

Ejemplo: Estudiar la continuidad de la función

$$f(x) = \sin x \text{ si } 0 \leq x, \quad f(x) = x^2 \text{ si } -1 < x < 0, \quad f(x) = -2x - 1 \text{ si } x \leq -1$$

```
[ > restart:
[ > with(plots) :
[ > g1:=plot(-2*x-1,x=-2..-1,color=green) :
[ > g2:=plot(x^2,x=-1..0,color=red) :
[ > g3:=plot(sin(x),x=0..2,color=blue) :
[ > display(g1,g2,g3) ;
```